Application Note AN3101-08: Two-Phase Quadrature Oscillator

by Shultz Wang

Introduction

This application note presents an algorithm for a simple sine wave generator. The generated sine waves may be outputted as pure test signals, used as inputs for other calculations, or used to control other functions.

Algorithm

Start off with the basic sum of angles formulae for sine and cosine.

$$\sin(A+\alpha) = \sin(A)\cos\alpha + \cos(A)\sin\alpha,$$
$$\cos(A+\alpha) = \cos(A)\cos\alpha - \sin(A)\sin\alpha.$$

The equations may be used as a predictor for the next point if the current point is known, by letting A be the current angle, and α be the step size.

$$\sin[n+1] = \sin[n]\cos\alpha + \cos[n]\sin\alpha,$$
$$\cos[n+1] = \cos[n]\cos\alpha - \sin[n]\sin\alpha.$$

This is sufficient information to generate a block diagram.



The incremental angle α is determined by the desired output frequency and is described by the following formula:

$$\alpha = 2\pi f/fs \text{ (f = selected frequency, fs = sampling frequency)}.$$

The oscillators need to be started by placing correct values of sin[n] and cos[n] in the registers. This is most easily done at one of the multiples of $\pi/2$ points. For convenience the point sin[n]=0, cos[n]=1 will be chosen.

Source Code

First calculate or select the following for each oscillator.

sinα
cosα
Ds (sine delay) Address (non-rotating)
Dc (cosine delay) Address (non-rotating)
Max (Maximum amplitude)

```
  ; Version 1: Explicit sine/cosine
CM     $(sinα)      $DcA  ; Dc*sinα
LCMA   $(cosα)      $DsA  ; Ds*cosα+Dc*sinα, store Ds in B
SCB    $(-sinα)     $DsA  ; Save new Ds, -Ds(sinα)
CMA    $(cosα)      $DcA  ; Dc*cosα-Ds*sinα
SCA    $040000      $DcA  ; Save new Dc
```

For low frequencies, cosα is approximately 1, and sinα is approximately α.

```
  ; Version 2: Approximate sine/cosine
  ; New Ds = Ds+Dc*α
  ; New Dc = Dc-Ds*α
CM     $(α)         $DcA  ; Dc*α
LCMA   $040000      $DsA  ; Ds+Dc*α, store Ds in B
SCB    $(-α)        $DsA  ; Save new Ds, -Ds*α
CMA    $040000      $DcA  ; Dc-Ds*α
SCA    $040000      $DcA  ; Save new Dc
```

Using the approximation introduces larger errors and increases THD. With sinα=$002000, cosα=$03ff7f, THD=0.0116%. With the approximation, α=$002000, THD=1.3%.

```
  ; Version 3: Approximate sine/cosine, one constant from memory
CM     $040000      $DcA  ; Dc
AMC    $0           $αA   ; Dc*α
LCMA   $040000      $DsA  ; Ds+Dc*α, store Ds in B
SCA    $040000      $DsA  ; Save new Ds
BMC    $0           $αA   ; Ds*α
CAM    $3c0000      $DcA  ; Dc-Ds*α
SCA    $040000      $DcA  ; Save new Dc
```

```
  ; Version 4: Two constants from memory
CM     $3c0000      $DsA  ; -Ds
AMC    $0           $sαA  ; -Ds*sinα
XCM    $040000      $cαA  ; cosα, store –Ds*sinα in B
AMB                 $DcA  ; Dc*cosα-Ds*sinα
LCA    $040000      $DcA  ; Store Dc in B
SCA    $040000      $DcA  ; Save new Dc
BMC    $0           $sαA  ; Dc*sinα
XCM    $040000      $cαA  ; cosα, store Dc*sinα in B
AMB                 $DsA  ; Ds*cosα+Dc*sinα
SCA    $040000      $DsA  ; Save new Ds
```

If there will be no microprocessor starting up the oscillators, there will need to be a startup routine placed right after the oscillator. Due to the resetting of the cosine, there may be a discontinuity in the generated cosine wave, therefore the sine wave should be preferentially used.

```
 ; Self reset for drift removal, auto startup; For versions 1, 2, 3
CM    $040000     $DsA  ; Load new Ds
SKIP  N           $4    ; Skip reset if new Ds < 0
CB    $040000           ; B->A
SKIP  Z           $1    ; Skip non-negative check if old Ds = 0,
                        ;     remove if auto startup not needed
SKIP  !N          $2    ; Skip reset if old Ds >= 0
DAC   $0          $(Max)      ; Load max amplitude
SCA   $0          $DcA  ; Dc=Max

 ; Version 4 self reset code addition; Place before oscillator
CM    $040000     $DsA  ; Read last Ds
SCA   $0          $TmA  ; Save last Ds
 ; Self reset for drift removal, auto startup; For version 4
SKIP  N           $5    ; Skip reset if new Ds < 0
CM    $040000     $TmA  ; Read old Ds
SKIP  Z           $1    ; Skip non-negative check if old Ds = 0, auto startup
SKIP  !N          $2    ; Skip reset if old Ds >= 0
DAC   $0          $(Max); Load max amplitude of output waveform
SCA   $0          $DcA  ; Dc=max
```

Sample oscillator:

```
 ; f=238.7Hz, Max=-12dB
CM    $002000     $400  ; Dc*α
LCMA  $03ff7f     $401  ; Ds+Dc*α, store Ds in B
SCB   $3fe000     $401  ; Save new Ds, -Ds*α
CMA   $03ff7f     $400  ; Dc-Ds*α
SCA   $040000     $400  ; Save new Dc
 ; Self reset for drift removal, auto startup
CM    $040000     $401  ; Load new Ds
SKIP  N           $5    ; Skip reset if new Ds < 0
CB    $040000           ; B->A
SKIP  Z           $1    ; Skip non-negative check if old Ds = 0, auto startup
SKIP  !N          $2    ; Skip reset if old Ds >= 0
DAC   $0          $010000     ; Load max amplitude
SCA   $0          $400  ; Dc=1
 ; Output sine wave
CM    $040000     $401  ; Read sine
SCA   $0          $410  ; Write sine to channel 0
```

NOTICE

Contact Information:

Wavefront Semiconductor
200 Scenic View Drive
Cumberland, RI 02864 U.S.A.
Tel: +1 401 658-3670
Fax: +1 401 658-3680
On the web at www.wavefrontsemi.com
Email: info@wavefrontsemi.com

Application note revised March, 2005